Lay 4.8

Applications to Difference Equations

Data:
Color Names

```
forestGreen = ColorData["Legacy", "ForestGreen"];
marsOrange = ColorData["Legacy", "MarsOrange"];
orange = ColorData["Legacy", "Orange"];
```

Visualizing a Sequence (Signal)

```
Clear[a, k];

a = Table[a[k] = (-3/4)^k, {k, 10}]
```

$$\left\{-\frac{3}{4}, \frac{9}{16}, -\frac{27}{64}, \frac{81}{256}, -\frac{243}{1024}, \frac{729}{4096}, -\frac{2187}{16384}, \frac{6561}{65536}, -\frac{19683}{262144}, \frac{59049}{1048576}\right\}$$

```
Show[
 dots = ListPlot[a,
   AxesLabel → {k, "a[k]"},
   AxesOrigin → {0, 0},
   PlotStyle → {Red, PointSize[Large]}],
 lines = ListPlot[a,
   PlotJoined → True,
   PlotStyle → orange]]
```



Lay 4.8, Example 2, *Linear Independence*

Show that three sequences are linearly independent.

```
row[k_] := {1^k, (-2)^k, 3^k};

casorati = {row[1], row[2], row[3]};
% // MatrixForm

RowReduce[casorati];
% // MatrixForm
```

$$\begin{pmatrix} 1 & -2 & 3 \\ 1 & 4 & 9 \\ 1 & -8 & 27 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Lay 4.8, Example 3, *Filtering Signals*

First Input Signal

```
Clear[as, ys, zs, ks, yData, zData];

as = {.35, .5, .35};

ys = Table[Cos[k π / 4],
          {k, 0, 12}];
ys = N[ys, 1]
```

```
{1., 0.7, 0, -0.7, -1., -0.7, 0, 0.7, 1., 0.7, 0, -0.7, -1.}
```

```
zs = Table[as.ys[[{k, k + 1, k + 2}]],
           {k, 1, 11}];
zs = N[zs // Chop, 1]
```

```
{0.703553, 0, -0.703553, -0.994975, -0.703553, 0, 0.703553, 0.994975, 0.703553, 0, -0.703553}
```

```
ks = Range[0, 10];
yData = Transpose[{ks, Drop[ys, -2]}];
zData = Transpose[{ks, zs}];

Show[
 yPlot = ListPlot[yData,
   Ticks → {2 Range[0, 6], Automatic},
   PlotJoined → True, AxesLabel → {k, None},
   PlotStyle → {PointSize[0.02], forestGreen}],
 zPlot = ListPlot[zData,
   PlotJoined → True,
   PlotStyle → {PointSize[0.02], marsOrange}]]
```



Conclusion : The output was shifted backward by one term.

Second Input Signal

```
Clear[as, higherFrequencyYs, zs, ks, yData, zData];

as = {√2/4, 1/2, √2/4};

higherFrequencyYs = Table[Cos[3 k π / 4],
                          {k, 0, 12}];
higherFrequencyYs = N[higherFrequencyYs, 1]
```

{1., −0.7, 0, 0.7, −1., 0.7, 0, −0.7, 1., −0.7, 0, 0.7, −1.}

```
newZs = Table[as.higherFrequencyYs[[{k, k + 1, k + 2}]],
           {k, 1, 11}];
newZs = N[newZs, 1]
```

$\{0. \times 10^{-2}, 0. \times 10^{-2}, 0. \times 10^{-2}, 0. \times 10^{-1}, 0. \times 10^{-2}, 0. \times 10^{-2}, 0. \times 10^{-2}, 0. \times 10^{-1}, 0. \times 10^{-2}, 0. \times 10^{-2}, 0. \times 10^{-2}\}$

```
ks = Range[0, 10];
yData = Transpose[{ks, Drop[higherFrequencyYs, -2]}];
zData = Transpose[{ks, newZs}];

Show[
 yPlot = ListPlot[yData,
   Ticks → {2 Range[0, 6], Automatic},
   PlotJoined → True, AxesLabel → {k, None},
   PlotStyle → {PointSize[0.02], forestGreen}],
 zPlot = ListPlot[zData,
   PlotJoined → True,
   PlotStyle → {PointSize[0.02], marsOrange}]]
```



Conclusion : This filter killed the higher frequency signal.

Lay 4.8, Example 4,
Solutions of a Homogeneous Difference Equation

Define the difference equation.

```
as = {1, -2, -5, 6};

ys = Table[y[k + j], {j, 0, 3}];

yeqn = as.Reverse[ys] == 0
```

```
6 y[k] - 5 y[1 + k] - 2 y[2 + k] + y[3 + k] == 0
```

Define the auxiliary equation, and find its roots.

```
rs = Table[r^k, {k, 0, 3}];

eqn = as.Reverse[rs] == 0

Factor[eqn]

soln = Solve[eqn, r] // Flatten
```

$6 - 5 r - 2 r^2 + r^3 == 0$

```
(-3 + r) (-1 + r) (2 + r) == 0
```

```
{r -> -2, r -> 1, r -> 3}
```

Check these roots.

```
{eqn /. r -> -2,
 eqn /. r -> 1,
 eqn /. r -> 3}
```

```
{True, True, True}
```

Lay 4.8, Example 5,
Solution Space of a Homogeneous Equation

Find a basis for the set of all solutions of our homogeneous equation.

```
yeqn

β = Table[r^k /. soln[[j]],
          {j, 3}]
```

```
6 y[k] - 5 y[1 + k] - 2 y[2 + k] + y[3 + k] == 0
```

```
{(-2)^k, 1, 3^k}
```

Lay 4.8, Example 6,
Non-homogeneous Difference Equation

Consider a non-homogeneous difference equation.

```
Clear[as, ys, r];

as = {1, -4, 3};

ys = Table[y[k + j], {j, 0, 2}];

yeqn = as.Reverse[ys] == -4 k
```

```
3 y[k] - 4 y[1 + k] + y[2 + k] == -4 k
```

Verify a particular solution.

```
y[k_] = k^2;

yeqn
% // Simplify
```

```
3 k^2 - 4 (1 + k)^2 + (2 + k)^2 == -4 k
```

```
True
```

Solve the associated homogeneous equation.
Define the auxiliary equation, and find its roots.

```
rs = Table[r^k, {k, 0, 2}];

eqn = as.Reverse[rs] == 0

Factor[eqn]

soln = Solve[eqn, r] // Flatten
```

$3 - 4 r + r^2 == 0$

$(-3 + r) (-1 + r) == 0$

{r → 1, r → 3}

Find a basis for the set of all solutions of the homogeneous equation.

```
β = Table[r^k /. soln[[j]],
        {j, 2}]
```

$\{1, 3^k\}$

Assemble the solutions to the non-homogeneous equation by adding
a single particular solution of the non-homogeneous equation to
the general solution of the homogeneous equation.

```
solution[k_, c1_, c2_] := y[k] + {c1, c2}.β

solution[k, c1, c2]
```

$c1 + 3^k c2 + k^2$

Lay 4.8, Example 7,
Reduction to Systems of First-Order Equations

Write the following equation as a first-order system.

```
Clear[x, y, k];

as = {1, -2, -5, 6};

ys = Table[y[k + j], {j, 0, 3}];

yeqn = as.Reverse[ys] == 0
```

```
6 y[k] - 5 y[1 + k] - 2 y[2 + k] + y[3 + k] == 0
```

Define x[k], and observe.

$$x[k\_] := \begin{pmatrix} y[k] \\ y[k+1] \\ y[k+2] \end{pmatrix};$$

$$a = \begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ -6 & 5 & 2 \end{pmatrix};$$

```
a.x[k] // MatrixForm
```

$$\begin{pmatrix} y[1+k] \\ y[2+k] \\ -6\,y[k] + 5\,y[1+k] + 2\,y[2+k] \end{pmatrix}$$

So, x[k + 1] == a.x[k].

```
Solve[yeqn, y[k + 3]] // Flatten
```

```
{y[3 + k] → -6 y[k] + 5 y[1 + k] + 2 y[2 + k]}
```